

// quart de cercle //

POUR quart_de_cercle
REPETE 45 [AVANCE 2 Tournedroite 2]
FIN

// Pétale = 2 quarts de cercle //

POUR pétale
REPETE 2 [quart_de_cercle Tournedroite 90]
FIN

// Fleur = 10 pétales //

POUR fleur
REPETE 10 [pétale Tournedroite 360/10]
FIN

// Plante = fleur + tige + pétale + tige //

POUR plante
COULEURTORTUE ROUGE

fleur
COULEURTORTUE VERTE

RECULE 130

pétale
RECULE 70

FIN

POUR fleurs

EFFACEECRAN ACCELERE

LEVECRAYON TOURNEGAUCHE 90 AVANCE 150

TOURNEDROITE 90 BAISSSECRAYON

REPETE 3 [

plante LEVECRAYON AVANCE 200 Tournedroite 90 AVANCE 150 TOURNEGAUCHE 90 BAISSSECRAYON]

FIN

// Taper dans la ligne de commande (fenêtre de travail):

fleurs

//

POUR peupliers

EFFACEECRAN CACHETORTUE ACCELERE

SEGMENT [-20 0] [-20 25] arbre 40 0 (-20) 25

SEGMENT [-60 0] [-60 25] arbre 40 0 (-60) 25

SEGMENT [-100 0] [-100 25] arbre 40 0 (-100) 25

SEGMENT [-140 0] [-140 25] arbre 40 0 (-140) 25

SEGMENT [20 0] [20 25] arbre 40 0 20 25

SEGMENT [60 0] [60 25] arbre 40 0 60 25

SEGMENT [100 0] [100 25] arbre 40 0 100 25

SEGMENT [140 0] [140 25] arbre 40 0 140 25

FIN

POUR arbre :lon_branche :cap_ :x :y

TESTE :lon_branche > 3

SIVRAI [

LEVECRAYON FIXEXY :x :y BAISSSECRAYON

FIXEHASARD 0 20

FIXECAP (:cap_ - 15) + (HASARD - 10)

AVANCE :lon_branche

arbre ((:lon_branche * 1) / 3) CAP XCOORD YCOORD

LEVECRAYON FIXEXY :x :y BAISSSECRAYON

FIXEHASARD 0 20

FIXECAP :cap_ + (HASARD - 10)

AVANCE :lon_branche

arbre ((:lon_branche * 4) / 5) CAP XCOORD YCOORD

LEVECRAYON FIXEXY :x :y BAISSSECRAYON

FIXEHASARD 0 20

FIXECAP (:cap_ + 15) + (HASARD - 10)

AVANCE :lon_branche

arbre ((:lon_branche * 1) / 3) CAP XCOORD YCOORD

]

RETOURNE
FIN

// Taper dans la ligne de commande (fenêtre de travail):
peupliers
//

POUR escargot
EFFACEECRAN BAISSSECRAYON CACHETORTUE
ACCELERE TRAIT 3 COULEURTORTUE NOIRE
// Tracé de la coquille. //
REPETE 550 [
AVANCE 0,01 * BOUCLE TOURNEDROITE 3
]
TOURNEGAUCHE 120 AVANCE 70 // Tracé du corps. //
TOURNEGAUCHE 20 AVANCE 20
REPETE 3 [
TOURNEDROITE 30 AVANCE 9
TOURNEDROITE 29 AVANCE 10
]
TOURNEDROITE 20 AVANCE 170
LEVECRAYON
RECULE 175
TOURNEDROITE 90 AVANCE 37
BAISSECRAYON
// Tracé des antennes. //
TOURNEDROITE 30 AVANCE 30 RECULE 30
TOURNEDROITE 15 AVANCE 30
LEVECRAYON VA [86 (-64)] BAISSSECRAYON VA [57 (-64)]
LEVECRAYON VA [-29 (-103)] BAISSSECRAYON VA [-159 (-103)]
LEVECRAYON VA [-159 (-103)] BAISSSECRAYON VA [-80 (-74)]
COLORIE [-21 (-10)] TILLEUL // on colorie la coquille. //
COLORIE [50 (-80)] ORANGE // on colorie le corps. //
COLORIE [-91 (-91)] ORANGE // on colorie la queue. //
TRAIT 1
FIN

// Taper dans la ligne de commande (fenêtre de travail):
escargot
//

POUR matrice
CACHETORTUE EFFACEECRAN
DONNE "y 0
REPETE 50 [
DONNE "x 0
REPETE 50 [
POINT [:y :x]
DONNE "x :x + 5
]
DONNE "y :y + 5
]
FIN

// Taper dans la ligne de commande (fenêtre de travail):
matrice
//

POUR cercle
DONNE "situ SITUATION
ELLIPSE [XCOORD YCOORD 35 35]
RAMENE SITUATION
FIN

POUR tore
EFFACEECRAN ACCELERE LEVECRAYON CACHETORTUE
REPETE 180 [cercle AVANCE 4 TOURNEDROITE 2]
FIN

// Taper dans la ligne de commande (fenêtre de travail):

```

tore
//

// Lorsque les équations différentielles se mettent à dessiner !
Voici une procédure "lente" //

POUR courbe1 :n
donne "couleurs [ blanche verte jaune rouge orange tilleul fuchsia orange marron rose violette argent grise noire bleue
olive ]
FIXEHASARD -280 280
EFFACEECRAN EFFACETEXTE CACHETORTUE
PARTAGE [ "x "y ]
DONNE "x [] DONNE "y []
poly 150 (360 / :n)
DONNE "h 0,05
REPETE 100 [ // 100 itérations seulement. //
EFFACETEXTE
COULEURTORTUE ITEM :couleurs 1+ RESTE BOUCLE 15
DONNE "xxm [] DONNE "yym []
REPETE :n [ // la courbe précédente poursuit la suivante.//
DONNE "mg ( RESTE BOUCLE :n ) + 1
DONNE "xxm PHRASE :xxm ( ITEM :x BOUCLE ) + ( :h * (( ITEM :x :mg ) - ITEM :x BOUCLE ))
DONNE "yym PHRASE :yym ( ITEM :y BOUCLE ) + ( :h * (( ITEM :y :mg ) - ITEM :y BOUCLE ))
SEGMENT [ ITEM :x :mg ITEM :y :mg ] [ ITEM :xxm BOUCLE ITEM :yym BOUCLE ]
// on trace ici. //
]
REPETE :n [ // on actualise la liste des nouvelles coordonnées.//
REPLACE :x ( ITEM : xxm BOUCLE ) [ BOUCLE ]
REPLACE :y ( ITEM : yym BOUCLE ) [ BOUCLE ]
]
]

FIN

// et une procédure "rapide" //

POUR courbe2 :n
donne "couleurs [ blanche verte jaune rouge orange tilleul fuchsia orange marron rose violette argent grise noire bleue
olive ]
FIXEHASARD -280 280
EFFACEECRAN EFFACETEXTE CACHETORTUE
PARTAGE [ "x "y ]
DONNE "x [] DONNE "y []
poly 150 (360 / :n)
DONNE "h 0,05
REPETE 300 [ // 300 itérations. //
EFFACETEXTE
COULEURTORTUE ITEM :couleurs 1+ RESTE BOUCLE 15
DONNE "xxm [] DONNE "yym []
//On calcule les nouvelles coordonnées des points extrêmes des listes :x et :y pour faciliter les
calculs. On peut à présent tronquer ces deux listes au fur et à mesure des itérations, d'où
accélération du tracé. //
DONNE "xd ( DERNIER :x ) + ( :h * ( ( PREMIER :x ) - ( DERNIER :x ) ) )
DONNE "yd ( DERNIER :y ) + ( :h * ( ( PREMIER :y ) - ( DERNIER :y ) ) )
DONNE "xxm PHRASE :xxm ( PREMIER :x ) + ( :h * (( ITEM :x 2 ) - ( PREMIER :x ) ) )
DONNE "yym PHRASE :yym ( PREMIER :y ) + ( :h * (( ITEM :y 2 ) - ( PREMIER :y ) ) )
SEGMENT [ ITEM :x 2 ITEM :y 2 ] [ DERNIER :xxm DERNIER :yym ]
SEGMENT [ :xd :yd ] [ DERNIER :xxm DERNIER :yym ]
DONNE "x SAUFPREMIER :x DONNE "y SAUFPREMIER :y
REPETE ( :n - 2 ) [ // la courbe précédente poursuit la suivante.//
DONNE "xxm PHRASE :xxm ( PREMIER :x ) + ( :h * (( ITEM :x 2 ) - ( PREMIER :x ) ) )
DONNE "yym PHRASE :yym ( PREMIER :y ) + ( :h * (( ITEM :y 2 ) - ( PREMIER :y ) ) )
SEGMENT [ ITEM :x 2 ITEM :y 2 ] [ DERNIER :xxm DERNIER :yym ] // on trace ici. //
DONNE "x SAUFPREMIER :x DONNE "y SAUFPREMIER :y
]
]
DONNE "xxm PHRASE :xxm :xd DONNE "yym PHRASE :yym :yd
DONNE "x [] DONNE "y []
REPETE :n [ // on actualise la liste des nouvelles coordonnées.//
DONNE "x PHRASE :x PREMIER :xxm
DONNE "y PHRASE :y PREMIER :yym
DONNE "xxm SAUFPREMIER :xxm

```

```
DONNE "yym SAUFPREMIER :yym
]
]
FIN
```

```
POUR poly :côté :angle
ECRISLIGNE [ Calcul des points initiaux. ]
DEROULE LEVECRAYON RECULE 100 Tournedroite 90
RECULE 50 TOURNEGAUCHE 90
LEVECRAYON
REPETE ENTIER ( 360 / :angle ) [
AVANCE :côté
TOURNEDROITE :angle
DONNE "x PHRASE :x XCOORD
DONNE "y PHRASE :y YCOORD
]
FIN
```

courbe1 5

```
// Taper dans la ligne de commande (fenêtre de travail):
courbe1 5
ou
courbe2 5
(le tracé de courbe1 est plus lent que celui de courbe2)
//
```

```
POUR serpent
EFFACEECRAN FIXEHASARD 100 17000
REPETE 10 [
COULEURTORTUE HASARD
TESTE CAP = 0
SIVRAI REPETE 60 [ AVANCE 1 Tournedroite 3 ]
SIFaux REPETE 60 [ AVANCE 1 TOURNEGAUCHE 3 ]
]
FIN
```